Numeric Recipe for Econometrics
Smoothing and Seasonal adjustment

M.Shimura
JCD02773@nifty.ne.jp

2007 年 3 月 24 日

目次

1 Moving Average

最初に移動平均を紹介したのは Poynting(1884) と言われる。

1.1 単純移動平均

J は、移動平均を次の簡潔なイディオムで表現できる。

mav=: +/\ % [

これは

mav=:(# %~ +/)\

でも表現できる。

経過と説明

J の infix（\) の機能は強力であり左引数（x）で指定した数で移動平均のためにずらした組み合わせを作る。

```
 ] s=. 10?.30
  6 27 9 16 28 19 26 10 20 14
```

3<\s NB. 3 **次の例**

```
+------+-------+-------+--------+-------+-------+--------+
|6 27 9|27 9 16|9 16 28|16 28 19|28 19 26|19 26 10|26 10 20|
+------+-------+-------+--------+-------+-------+--------+..
```

```
   3+/\s  NB. add
42 52 53 63 73 55 56 44
```

これを n で割れば単純移動平均が求められる。

```
   3 mav s NB. divide by 3
14 17.3333 17.6667 21 24.3333 18.3333 18.6667 14.6667
```

中心化

移動平均は, 次数（左パラメーター x) が偶数の場合には, 原データと整合させるために中心化が行われる。mav_c は次数の奇数、偶数を自動判別して移動平均を行う。

```
mav=:+/\ % [
mav_c=: 2&mav@mav ‘ mav @. (2&|@[)   NB.centred MA


   s=.10?.20
  a , (2 mav a),: 2 mav_c a


 6  3     19     15  10   14   0   7   12 17
  4.5 11    17  12.5  12     7 3.5 9.5 14.5  0
  0 7.75 14 14.75 12.25 9.5 5.25 6.5  12   0
```

1.1.1 12 カ月移動平均

$mav12$ は mav の引数を１２に固定して組込み中心化を行っている。

```
mav=: +/\ % [
NB. Season adjustment using 12 month

mav12=: [: 2&mav 12&mav

      2<\ 12 mav i. 24
+-------+-------+-------+-------+--------+---------+---------+
|5.5 6.5|6.5 7.5|7.5 8.5|8.5 9.5|9.5 10.5|10.5 11.5|11.5 12.5|
+-------+-------+-------+-------+--------+---------+---------+

   mav12 i. 24
6 7 8 9 10 11 12 13 14 15 16 17
```

12 項移動平均は 13 項を取って最初と最後の項に $\frac{1}{2}$ のウエイトを置いた
$(\frac{1}{2} 1\ 1\ 1\ 1\ 1\ 1\ 1\ 1\ 1\ 1\ 1\ \frac{1}{2})$ １３項加重移動平均と同じである。

1.2　加重移動平均

移動平均の項にウエイトを置く方法。

1.2.1　3 項加重移動平均

３ヶ月の移動平均値を更に移動平均する場合。一層のスムージングがはかれる。

$$Q = \frac{\frac{Q_{t-2}+Q_{t-1}+Q_t}{3}\ \frac{Q_{t-1}+Q_t+Q_{t+1}}{3}\ \frac{Q_t+Q_{t+1}+q_{t+2}}{3}}{3}$$
$$= \frac{1}{9}Q_{t-2} + \frac{2}{9}Q_{t-1} + \frac{3}{9}Q_t + \frac{2}{9}Q_{t+1} + \frac{1}{9}Q_{t+2}$$
$$= \frac{1}{9}[1\quad 2\quad 3\quad 2\quad 1]$$

これは、上記のウエイトによる５ヶ月の加重移動平均と同じである。四季の変動の特徴をよく捕らえることができる。

1.2.2　補完

加重 3 月移動平均の欠項は, 次のウェイトを用いて両端を補完することができる。

$$Q_{t+1} = \frac{1}{9}Q_{t+3} \rightarrow \frac{1}{9}\left(\frac{Q_t+Q_{t+1}+Q_{t+2}}{3}\right)$$
$$Q_{t+1} = \frac{3\quad 7\quad 10\quad 7}{27}$$
$$Q_{t+2} = \frac{2}{9}Q_{t+3} \rightarrow \frac{2}{9}\left(\frac{Q_t+Q_{t+1}+Q_{t+2}}{3}\right) + \frac{1}{9}\left(\frac{Q_t+Q_{t+1}+Q_{t+2}}{3}\right)$$
$$Q_{t+2} = \frac{2\quad 3\quad 4}{9}$$

cpl_mavw3 は両端の補完を行う。plot_w3 は cpl_mavw3 を組み込んだ.

1.2.3　Spencer 移動平均

スペンサー１５ポイント移動平均は U.S. 商務省センサス局法 X-10 で用いられた。
ウエイトを次のように置いており、隣接に重心をおくので、三角形や低位の数値が破壊されないでよく残っている。

これは、次のウエイトに依っている。

$$\frac{1}{4}\left[\begin{array}{ccccc} -3 & 3 & 4 & 3 & -3 \end{array}\right]$$
$$\frac{1}{5}\left[\begin{array}{ccccc} 1 & 1 & 1 & 1 & 1 \end{array}\right]$$
$$\frac{1}{4}\left[\begin{array}{ccccc} -3 & 3 & 4 & 3 & -3 \end{array}\right]$$
$$\frac{1}{4}\left[\begin{array}{ccccc} -3 & 3 & 4 & 3 & -3 \end{array}\right]$$

これは、実際には、次のウエイトになる。**[?]** **[?]**

$$a_0 = \frac{74}{320}, a_{\pm 1} = \frac{67}{320}, a_{\pm 2} = \frac{46}{320}, a_{\pm 3} = \frac{21}{320},$$
$$a_{\pm 4} = \frac{3}{320}, a_{\pm 5} = \frac{-5}{320}, a_{\pm 6} = \frac{-6}{320}, a_{\pm 7} = \frac{-3}{320}$$

Spencer 移動平均のウエイトは 15 項限定であり、加重１５項移動平均のパラメーターと一致する。



図 1　Stock Price of IBM NY Market/Spencer

1.2.4 Script

```
spencer2=: 3 : 0
WEIGHT=: _3 _6 _5 3 21 46 67 74 67 46 21 3 _5 _6 _3 % 320
 y,. ; WEIGHT +/ . * (L:0)  15<\ ( (7 # {.),] ,7 #{:) y
)
```

C.Reiter によるスマートな Tacit のスクリプト

```
NB. ===============================
NB. C.Reiter Fractals Visuarization and J (2nd Edition 2000)
NB. ==============================
wts=: (|.,}.) 74 67 46 21 3 _5 _6 _3%320
locspen=: (+/ . *)&wts
spencer=: 15"_ locspen\ (7: # {.),] ,7:#{:
```

1.2.5 Henderson 移動平均

U.S. 商務省センサス局法 X-11,X-12 の採用する季節調整法に採り入れられている手法で、基本的には、全変動を、Trendcycle(TC)、季節変動 (S)、不規則変動 (I) の３つの成分に分けるときの、TC を求めるのに使用される重み付き移動平均法である。

$$H^n(B) = \sum_{i=-(n-1)/2}^{(n-2)/2} h_i^{(n)} B^i, \quad i = 0, \pm1, \pm2, \cdots, \pm(n-1)/2$$

B , 時間の遅れ、　F、時間の前のパラメータ

B is B(back) F is F(Foward)

$h_i^{(n)}$重み係数　$m = (n+3)/2$

$$h_i^{(n)} = \frac{315((m-1)^2 - i^2)(m^2 - i^2)((m+1)^2 - i^2)((3m^2 - 16) - 11i^2)}{8m(m^2-1)(4m^2-1)(4m^2-9)(4m^2-25)}$$

５項のウエイト

$$H^{(5)}(B) = -\frac{21}{286}B^2 + \frac{84}{286}B + \frac{160}{286} + \frac{84}{286}F - \frac{21}{286}F^2$$

10mm

７項のウエイト

$$H^{(7)}(B) = -\frac{42}{715}B^3 + \frac{42}{715}B^2 + \frac{210}{715}B + \frac{295}{715} + \frac{210}{715}F + \frac{42}{715}F^2 - \frac{42}{715}F^3$$

Spencer や Henderson の移動平均は、原系列を追いながら滑らかさを実現していることである。Henderson の７項は、原系列を追っていくが、１５項では滑らかになる。Spencer の１５項は、Henderson の１５項と図で表せば、概ね同じような滑らかさを持っており、ウエイトの差は微小である。



図２　Stock Price of IBM NY Market
Spencer,Henderson

IBM Stock Price of NY Market の最初の１００ポイントを Spencer と Henderson の１５ポイントで plot してみると、ほとんど線が重なる。henderson は項数が任意に取れるところが優れている。

交通量やデパート売り上げのデータでは、８月や１２月のピークが大きいが、henderson は高次になるほど、追従よりも滑らかさを求めていく保守的 (Concerbative) な振舞いがあると言われる。

1.2.6　Script

スクリプトは竹内による。 [**?**]　 henderson の移動平均の項数は任意（奇数）で、１３－１５項では Spencer と極めて類似した結果となる。

```
    henderson=:4 : 0
NB. Usage: i.e. 15&henderson y  //or 7 henderson y
wt=: x  hnwgt i:-:<:x
x (+/ . *)&wt\((<.@-:x)#{.y),y,(<.@-:x)#{:y
)
   hnwgt=:4 : 0
m=: -: x +3 NB. (n+3)/2
nm=: 315*(((m-1)^2)-*:y)*((m^2)-*:y)*(((m+1)^2)-*:y)*(((3**:m)-16)-11**:y)
hi=:nm%8*m*((m^2)-1)*((4**:m)-1)*((4**:m)-9)*((4**:m)-25)
)
```

1.2.7　**移動平均の** PLOT

移動平均を求め原データと同時にグラフにあらわす関数を用意した.

・任意の次数（例では３＝四半期）と原系列を plot する

偶数の場合はグラフにあらわすため中心化の処理を行っている。

plot_wmav	weighted mav	plot_wmav n

$plot\_mav12$	12 months mav	plot_mav12 n
$plot\_mav12cpl$	12 months mav	plot_mav12cpl n
$plot\_w13$	13 weighted mav and mav12cpl	plot_w13 n
$plot\_spencer$	spencer	plot_spencer n
$plot\_henderson$	henderson 15 項	plot_henderson n
$plot\_henderson2$	henderson 任意の項 and spencer	m. plot_henderson2 n
$plot\_comb\_all$	henderson 任意の項と and spencer ,mav12cpl	m. plot_comb_all n henderson は次数は奇数のみ 31 33 付近は中心化１２月移動平均と類似　両端は計算方法の差によるばらつきがある。

1.3 Script List

mav	Moving Average	m mav n
mav_c	Centred mav auto classified	m mav_c n
$mav12$	12 month mav centred	mav12 n
$spencer$	spencer's 15 point moving average	spencer n
$henderson$	henderson n point moving average	m (次数) henderson n 次数は奇数を指定

2　指数平滑化

2.1　指数平滑化

時系列データの平滑化で、計算が簡単でよく使い込まれた便利な方法に指数平滑化がある。任意の $0 < w < 1$ のウェイトを与えるとウェイトにより平滑化を行う。最初の方の値の影響がが後まで残る方法であるが、ウエイトを決める方法はないので、グラフを見ながら適当な値を捜す。

$$x_1^- = x_1$$
$$x_2^- = wx_1{}^- + (1-w)x_2$$
$$x_3^- = wx_2{}^- + (1-w)x_3 = w^2x_1 + w(1-w)x_2 + (1-w)x_3$$

Working Example

0.2 smexp s2

2.1.1　Script

```
NB. written by Giichiro Suzuki
   smexp=:4 : 0
s=. {.y=. y
 while. 1<#y do.
  s=.s,(x *{:s)+(1-x )*{.y=.}.y
 end.
)
```

2.2　ホルトの 2 段階指数平滑化

Holt's two-parameter double exponential smoothing model

$S_1 = \alpha Y_t + (1-\alpha)(S_{t-1} + bt - 1)$



図 3　東京の冬の気温と指数平滑化

$$b_t = \beta(S_1 - S_{t-1}) + (1-\beta)b_{t-1}$$
$$F_{t+m} = S_1 + b_t m$$

	Starting base
α = level smoothing constant	
S_1 = smoothed at end of period t	Y_1
β = trend smoothing constant	
b_1 = smoothed trend in period t	$(Y_2 - Y_1 + Y_4 - Y_3)/2$
m = forecast horizon	

α, β は一意には定まらない。この例では、$\alpha = 0.2$ $\beta = 0.6$ が最良であるが、周辺の数字を入れ替えてトライして、残差が最少のものを選択して、モデルとする。

```
0.2 0.8  holt DATL0

(1)    (2)     (3)       (4)
data  ANS_S   ANS_B     Fitted
```

	0.2 0.8	holt	DATL
63	63	_0.5	0
64	62.8	_0.26	62.5
66	63.232	0.2936	62.54
64	63.6205	0.369504	63.5256
68	64.792	1.01111	63.99
70	66.6425	1.68261	65.8031
67	68.0601	1.4706	68.3251
69	69.4245	1.38569	69.5307
74	71.4482	1.89606	70.8102
72	73.0754	1.68098	73.3442
73	74.4051	1.39996	74.7564
78	76.244	1.75115	75.8051
77	77.7962	1.59192	77.9952
80	79.5105	1.68983	79.3881
83	81.5602	1.97778	81.2003
81	83.0304	1.5717	83.538
86	84.8817	1.79536	84.6021
88	86.9416	2.00703	86.6771
85	88.1589	1.37525	88.9487
87	89.0273	0.969777	89.5342
92	90.3977	1.29024	89.9971
95	92.3503	1.82017	91.6879
93	93.9364	1.63288	94.1705
97	95.8554	1.8618	95.5693
99	97.9738	2.06704	97.7172
96	99.2327	1.42051	100.041
102	100.923	1.636	100.653
101	102.247	1.38663	102.559
103	103.507	1.28528	103.633
105	104.834	1.31855	104.792

(2) Smoothed Level

(3) Smoothed Trend

$$\mathrm{RSE} = \sqrt{\sum \frac{e^2}{n-1}}$$



図 4　holt's 2 parameter trend model

2.2.1　残差分析の自動計算

残差計算が大変なので、1に近いパラメーターを与えると、0まで 0.05 ピッチで 0 までぐるぐる回し、最小値を選択して、ホルト法を計算するルーチンを作った。結果は降順にソートしてあるので最下行をとればよい。(初期値によって結果が多少異なる)

```
0.2 0.8  holt_loop DATL


 0.2 0.45 2.46177
0.15  0.7 2.41542
 0.2  0.5 2.39981
0.15 0.75 2.37576
 0.2 0.55 2.35378
0.15  0.8 2.34396
```

```
0.2  0.6  2.3198
0.2 0.65 2.29505
0.2  0.7 2.27745
0.2 0.75 2.26552
0.2  0.8 2.25821    NB. minimum residual
```

```
plot |: }. 0 3 {("1)  0.2 0.8  holt DATL
pd 'eps holt0.eps'
```

指数平滑化	smexp	m smexp n
Holt の 2 段階指数平滑化	holt hol_loop	m_1 m_2 holt n holt_loop n

3 Seasonal Adjustment

季節調整法は経験の積重ねの内で発展してきたもので、絶対的な理論が存する訳ではない。

Pindyck and Rubinfeld の Econometric models and Forcast(1998) に目を通していたら、次のような記述があった。

Seasonal adjustment techniques are basically ad hoc methods of computing seasonal indices. National economic data in the United States usually are seasonal adjusted by the Census II method,which was developed by the Bureau of the Census of the U.S.Departmrnt of Commerce. The Census II method is a rather detailed and complicated procedure (and is amazingly ad hoc),

各種の移動平均を５回以上駆使する X12 は歌舞伎の隈のような厚塗りであり、大型研磨機で鉛筆を削るようなものである。

Pindic & Rubinfeld は次の簡潔な方法を紹介している。

Additive $Y = L + S + C + I$

Multiplicative $Y = L \times S \times C \times I$

L value of the long-term secular trend in series

S value of seasonal component

C (long term) cyclical component

I irregular component

ここで S を取り除くには、先ず単純移動平均により $L \times C$ を推計し

$$\hat{y} = \frac{1}{12}(y_{t+6} + \cdots y_t + y_{t-1} + \cdots y_{t-5})$$

オリジナルデータを $L \times C$ で割ると $S \times I$ が求められる。

$$\frac{L \times S \times C \times I}{L \times C} = S \times I = \frac{y_t}{\hat{y}} = z_i$$

I を求める

$$\hat{z}_1 = \frac{1}{4}(z_1 + z_{13} + z_{25} + z_{37})$$

$$\hat{z}_2 = \frac{1}{4}(z_2 + z_{14} + z_{26} + z_{38})$$

$\cdots\cdots$

$$\hat{z}_{12} = \frac{1}{4}(z_{12} + z_{24} + z_{36} + z_{48})$$

trend and cyclical(TC) components are estimated using moving averages

3.1　四季の調整

Working Example

DeLurgio の BIGQTR のデータと計算メソッドに従って、季節データを季節変動指数を算出して、TCe を求め、更に、時間との単回帰により、トレンド T を求め、グラフに表してみよう。

この例では中心化４項移動平均で S を求めている。データは四季単位である。

Year	Quarter	Sales
1	1	72
1	2	110
1	3	117
1	4	172
2	1	76
2	2	112
2	3	130
2	4	194
3	1	78
3	2	119
3	3	128
3	4	201
4	1	81
4	2	134
4	3	141
4	4	216

４季の変動係数を求める

```
  adj_4season_sub a
0.606313 0.919069 0.992121 1.4825
```

原データを変動係数で割ると deseasonal が求められる。

Deseasonal を一次の最少自乗法で回帰して Trend を求め、これを変動係数で割り戻

すと推計値が得られる。
　ここではトレンドを１次式で回帰しているが、ここを低次の多項式などに変更すればさらに詳細に分解できる。

```
   adj_4season DAT

Y  seasonal deseasonal Trend Predict
---------------------------
 72 0.606313 118.751 115.554 70.0621
110 0.919069 119.686 117.409 107.907
117 0.992121 117.929 119.264 118.324
172   1.4825  116.02 121.118 179.558
 76 0.606313 125.348 122.973 74.5601
112 0.919069 121.863 124.828 114.725
130 0.992121 131.032 126.682 125.684
194   1.4825  130.86 128.537 190.556
 78 0.606313 128.646 130.392 79.0581
119 0.919069 129.479 132.246 121.543
128 0.992121 129.016 134.101 133.044
201   1.4825 135.582 135.955 201.554
 81 0.606313 133.594  137.81  83.556
134 0.919069   145.8 139.665 128.361
141 0.992121  142.12 141.519 140.404
216   1.4825   145.7 143.374 212.552


 plot |:  adj_4season a
```

mav_c	移動平均の標準型。偶数では中心化を行う。	m. mav_c n.
adj_4season adj_4season_sub	４季の変動指数を求め、回帰とBest Fitting	adj_4season n



図 5　季節データと TCe

3.1.1　Script

```
adj_4season_sub=: 3 : 0
NB. find season_index
TMP1=.(2}._2}. y) ,. 4&mav_c Y0=: (-4|#y)}. y
SEASON0=. 2|. "1 >_4<\ %/"1  TMP1
SEASON1=. +/ SEASON0 % (# TMP1) % 4
SEASON1 * 4 % +/ SEASON1
)

adj_4season=: 3 : 0
NB. using mav_c and reg0
Y0=. (-4|#y)}. y
SEASON=. adj_season_sub Y0
TMP0=. Y0,.;(_4<\Y0) % L:0 SEASON NB. const data
f=. reg0 {:"1 TMP0 NB. reg /find ternd
TMP1=. TMP0,. (1,.>: i. # TMP0) +/ . * f NB. reg_t use >: i. # y
```

```
TMP1,. ;SEASON * (L:0) _4<\ 2{"1 TMP1
)
```

Working Example GDP(1994-2007 Real) の deseasonalize

LSCI から S を除くと LCI が残りこれを deseason とする。ここから 1 次回帰で T を求めると T の周りで CI が変動する。T を S で割り戻したものが fitted である。

右は加重 deseason(red) と 3 項移動平均 (green) との比較。

(1994 1) plot_4season GDP



図 6　GDP(Japan 1994-2007 real)



図　7　GDP(real)=blue　sdeseason=red wma3=green

3.2　12 月の deseason

移動平均法により季節変動指数を求め、季節変動を除去することで、トレンドをもとめる。

3.2.1　季節変動指数

12 月の時系列データが短い時は全期間で季節変動指数を求める.

旧通商産業省の統計マニュアル「指数の作成と利用（１９９９)」には、直近の５年をとり最大値と最小値を除いた３年分で、季節変動指数を求める手法が紹介されている。特異値が除かれて便利なので、この方法も紹介する。　係数の合計を１２と微調整して最終の季節変動指数とする。

Working Example Delurgio Seriesf

Delurgio の seriesf のデータとプロセスを再現する。

```
 5j2 ":   (T1 % T0)
 1.21 1.15 1.02 0.95 0.90 0.81 0.84 0.93 0.95 1.02 1.09 1.14
 1.20 1.10 1.04 0.95 0.88 0.82 0.91 0.96 1.02 1.03 1.12 1.14
 1.10 1.06 0.96 0.89 0.83 0.91 0.94 0.95 1.03 1.05 1.12 1.19
```

```
 5j2 ":  (+/%#)  (T1 % T0) NB. sum each month
 1.17 1.10 1.01 0.93 0.87 0.85 0.89 0.95 1.00 1.04 1.11 1.15

 +/  (+/%#)  (T1 % T0) NB. sum whole year
12.0701

12%  +/  (+/%#)  (T1 % T0)NB. adjust to 12
0.994196

5j2 ": 6|. tmp* 12%  +/  (+/%#)  (T1 % T0) NB. rotate 6
 0.89 0.94 0.99 1.03 1.11 1.15 1.17 1.10 1.00 0.93 0.86 0.84

 5j2 ":  saj DF
 0.89 0.94 0.99 1.03 1.11 1.15 1.17 1.10 1.00 0.93 0.86 0.84
```

3.3　Trend とＯＬＳによる回帰

季節変動指数 (2) が求まると、データから季節変動を除して取り除き、 最少自乗法による単回帰で Trend を求め、季節変動指数を乗じて、Fitted Value を求めている。

```
     24{. adj_season DF
data deseason trend  fitted
----------------------------
546 615.234 663.859 589.153
578 614.084 670.861 631.441
660 664.795 677.863 672.974
707 686.886 684.865  704.92
```

738	667.381	691.867	765.077
781	680.371	698.869	802.233
848	727.773	705.87	822.479
818	745.662	712.872	782.029
729	729.111	719.874	719.765
691	746.325	726.876	672.993
658	761.132	733.878	634.439
604	716.424	740.88	624.618
629	708.759	747.882	663.721
711	755.387	754.884	710.527
729	734.296	761.886	756.391
798	775.297	768.888	791.404
861	778.611	775.89	857.991
903	786.652	782.892	898.684
968	830.76	789.894	920.383
894	814.942	796.896	874.204
860	860.13	803.898	803.776
792	855.412	810.9	750.788
739	854.828	817.902	707.077
699	829.107	824.904	695.457

....



図 8　季節変動指数



図 9　Trend,TredCyclee

Working Example

デパートの売上高（店舗調整なし、1997/2 - 2007/1) の deseason

データの出典　東洋経済統計月報

```
(1997 2)  plot_season ;{."1 DP
```



図 10　$deseason = red, Tred = green$

```
saj=:3 : 0
NB. using whole manthes to calc
NB. last update 30 June 2005 /tmp2 add <.
NB. Season Adjusument using Moving Average
NB. Usage: saj y  // y  is data ( ex  5 years=60 months)
T0=: (TMP1=:(|(# TMP0)%12 ) , 12) $  TMP0=: mav12 y
T1=: TMP1 $   6}. _6}. y
6 |. T3* 12 %  +/ T3=:(+/%#) T1%T0
)


adj_season=: 3 : 0
NB. 12 season data //using centered moving average
```

```
Y0=. (-(12| # y )) }. y  NB. drop out of 4 times from last //if 0 drop nothing
S40=.( 6}. _6}. Y0),. mav12 Y0
S41=. 6|."1 >_12<\ %/ "1 S40    NB. _4 non overlap,2|. is rotate
S42=.+/ S41 % ({. $ S40) % 12
SEASON=:S42 * 12 %(+/ S42)                     NB. Sasonal
S44=:Y0,. ,> (_12 <\ Y0) % L:0 SEASON      NB. Deseasolized: TCe
f=: (;{:"1 S44) %. X=.1 ,. >: i. # S44 NB. f with OLS
TREND=: X +/ . * f
S44 ,.TREND,.;(_12<\TREND)* L:0 SEASON              NB. trend
)
```

saj	季節変動指数を求める.whole months	saj n.(List)
saj2	季節変動指数を求める。last 3 years	saj2 n.(List),usualy using saj2
adj_season	季節変動指数を求め、回帰と Best Fitting を行う	adj_season n.(List)
adj_season2	季節変動指数を求め、回帰と Best Fitting を行う.saj2	adj_season2 n.(List),usualy using adj_season2
adj_season3	季節変動指数を求め、回帰と Best Fitting を行う.saj3	adj_season3 n.(List)
adj_season_ln	季節変動指数を求め、回帰と Best Fitting を行う (対数回帰)	adj_season_ln n.(List)
plot_adj_season		
loop_season	季節調整を行った後、対数により Erastically を求める	loop_season n(matrix many data OK)

3.3.1 参考文献

Stepaen A. Delurgio Forcasting Principles and Applications McGraw-Hill 1998

Gareth Janacek Practical Time series Arnord 2001

R.S.Pingyck D.L.Rubinfeld Economic Models and Economic Forcasts (4th.edition) McGraw-Hill 1998

C.Reiter [Fractal Visualization ans J 2nd ed.]

鈴木義一郎　J 言語による統計分析　森北出版　1996